
	Reason for change:

	At test step 3 SS transmits TRACKING AREA UPDATE REJECT with cause as ''UE implicitly detached”. Later SS expect Attach request from the UE on the new RRC Connection. However core specification does not indicate if UE should wait for the release of existing RRC Connection or send the Attach Request on the existing connection.

	A timer t_WaitForAttach for 1 second is introduced. Handling of ATTACH REQUEST is added in both scenarios i.e. ATTACH REQUEST is received on the existing RRC Connection within 1 sec or it is received after the RRC Connection has been released.

ATTACH REQUEST is received as Integrity protected on the existing RRC Connection.

1.1 Change 1
	Testcase name
	9.2.3.1.14

	Protocol
	EMM

	Function Name
	fl_TC_9_2_3_1_14_Body()

	Reason for change
	At test step 4, after sending TRACKING AREA UPDATE REJECT message SS is releasing the RRC connection, however specs allow to send the Attach Request on the existing RRC Connection. A conformant UE may send Attach Request before RRC Connection Release and that may leads to the failure of the test case.
Also in case ATTACH REQUEST is received on the existing RRC Connection it will be received as Integrity Protected.

Note: Prose CR will be raised at next RAN5 meeting. Similar changes were agreed for test case 9.2.3.1.13 at RAN5#49 through CR R5-106114


	Summary of change
	A timer t_WaitForAttach for 1 second is introduced. Handling of ATTACH REQUEST is added in both scenarios i.e. ATTACH REQUEST is received on the existing RRC Connection within 1 sec or it is received after the RRC Connection has been released.
ATTACH REQUEST is received as Integrity protected on the existing RRC Connection.

	Source of change
	NAS_TrackingArea.ttcn


Before:
	  function fl_TC_9_2_3_1_14_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NAS_MSG_Indication_Type v_NasInd;

    var SRB_COMMON_IND v_ReceivedAsp;

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellA );

    var template PTMSI_Signature v_PTMSISignature := omit;

    timer t_Wait := f_EUTRA_SetTimerToleranceMax (eutra_CellB, nasTTCNTimer, 30.0 ); // R5-106675

    if (pc_UTRA or pc_GERAN) {

      v_PTMSISignature := cr_PTMSI_Signature (px_PTMSI_SigDef);

    }

    //@siclog "Step 1" siclog@

    //+   Set the cell type of cell A to the ''Non-Suitable cell''. Set the cell type of cell B to the ''Serving cell''.

    f_EUTRA_SetCellPower(eutra_CellB, tsc_ServingCellRS_EPRE, cs_TimingInfo_Now);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE, cs_TimingInfo_Now);

    //@siclog "Step 2" siclog@

    //+  The UE transmits a TRACKING AREA UPDATE REQUEST on Cell B.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_OldGutiPTMSISig(v_UpdateType,

                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),

                                                                                                 v_PTMSISignature)));

    //@siclog "Step 3" siclog@

    //+  The SS transmits a TRACKING AREA UPDATE REJECT message with EMM cause = ''UE implicitly detached'' as specified.

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_EmmCause_ImplicitlyDetached))));

    f_EUTRA_RRC_ConnectionRelease_ImmediateReset(eutra_CellB, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other));

    //@siclog "Step 4" siclog@

    //+  Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds on cell B? [1,P]

    t_Wait.start;

    alt {

      [] SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellB,

                                         cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))

        {

          t_Wait.stop;

          f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellB, cs_TimingInfo_Now, tsc_RRC_TI_Def);

          SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellB,

                                             cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                             cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                               cdr_ATTACH_REQUEST_WithGutiAndPTMSI(v_AttachType,

                                                                                                                   v_AdditionalUpdateType,

                                                                                                                   f_EUTRA_SecurityKSIasme_Get(),

                                                                                                                   f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),

                                                                                                                   cr_TAI(v_PLMN, bit2oct(v_Tac)),

                                                                                                                   v_PTMSISignature),

                                                                               cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, f_GetPdnType()))))

            -> value v_ReceivedAsp;

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step 4");

        }

      [] t_Wait.timeout

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "step 4");

        }

    }
    //@siclog "Step 5 - 16" siclog@

    f_EUTRA_IdleUpdated_Step5_17(eutra_CellB, STATE2_IDLEUPDATE, NORMAL, v_ReceivedAsp.Signalling.Nas[0], false);

  } // function fl_TC_9_2_3_1_14_Body


After:
	  function fl_TC_9_2_3_1_14_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NAS_MSG_Indication_Type v_NasInd;

    var SRB_COMMON_IND v_ReceivedAsp;

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellA );

    var template PTMSI_Signature v_PTMSISignature := omit;

    timer t_Wait := f_EUTRA_SetTimerToleranceMax (eutra_CellB, nasTTCNTimer, 30.0 ); // R5-106675

    timer t_WaitforAttach := f_EUTRA_SetTimerToleranceMax( eutra_CellB, nonProtocolTimer, 1.0);    

    if (pc_UTRA or pc_GERAN) {

      v_PTMSISignature := cr_PTMSI_Signature (px_PTMSI_SigDef);

    }

    //@siclog "Step 1" siclog@

    //+   Set the cell type of cell A to the ''Non-Suitable cell''. Set the cell type of cell B to the ''Serving cell''.

    f_EUTRA_SetCellPower(eutra_CellB, tsc_ServingCellRS_EPRE, cs_TimingInfo_Now);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE, cs_TimingInfo_Now);

    //@siclog "Step 2" siclog@

    //+  The UE transmits a TRACKING AREA UPDATE REQUEST on Cell B.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_OldGutiPTMSISig(v_UpdateType,

                                                                                                 f_EUTRA_SecurityKSIasme_Get(),

                                                                                                 v_AdditionalUpdateType,

                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),

                                                                                                 v_PTMSISignature)));

    //@siclog "Step 3" siclog@

    //+  The SS transmits a TRACKING AREA UPDATE REJECT message with EMM cause = ''UE implicitly detached'' as specified.

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_EmmCause_ImplicitlyDetached))));

    //@siclog "Step 4" siclog@

    //+  Check: Does the UE transmit an ATTACH REQUEST message in the next 1second 

    t_WaitforAttach.start; // R5-106114

    alt {

      []  SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,

                                             cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected, //tsc_SHT_NoSecurityProtection, 

                                                                               cdr_ATTACH_REQUEST_WithGutiAndPTMSI(v_AttachType,

                                                                                                                   v_AdditionalUpdateType,

                                                                                                                   f_EUTRA_SecurityKSIasme_Get(),

                                                                                                                   f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),

                                                                                                                   cr_TAI(v_PLMN, bit2oct(v_Tac)),

                                                                                                                   v_PTMSISignature),

                                                                               cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, f_GetPdnType()))))

            -> value v_ReceivedAsp

          {

            t_WaitforAttach.stop;

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step 4");

          }

      [] t_WaitforAttach.timeout

        {

          //@siclog "Step 3a" siclog@

          //+   The SS releases the RRC connection.

          f_EUTRA_RRC_ConnectionRelease_ImmediateReset(eutra_CellB, cs_RRCConnectionRelease(tsc_RRC_TI_Def, other));

          //Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds?

          //@siclog "Step 4" siclog@

          t_Wait.start;

          alt {

            [] SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellB, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))

              {

                t_Wait.stop;

                f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellB, cs_TimingInfo_Now, tsc_RRC_TI_Def);

                SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellB,

                                                 cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                 cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected, //tsc_SHT_NoSecurityProtection, 

                                                                                   cdr_ATTACH_REQUEST_WithGutiAndPTMSI(v_AttachType,

                                                                                                                       v_AdditionalUpdateType,

                                                                                                                       f_EUTRA_SecurityKSIasme_Get(),

                                                                                                                       f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),

                                                                                                                       cr_TAI(v_PLMN, bit2oct(v_Tac)),

                                                                                                                       v_PTMSISignature),

                                                                                   cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, f_GetPdnType()))))

                  -> value v_ReceivedAsp;

                f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step 4");

              }

            [] t_Wait.timeout

              {

                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "step 4");

              }

         }

        }

    }
    //@siclog "Step 5 - 16" siclog@

    f_EUTRA_IdleUpdated_Step5_17(eutra_CellB, STATE2_IDLEUPDATE, NORMAL, v_ReceivedAsp.Signalling.Nas[0], false);

  } // function fl_TC_9_2_3_1_14_Body

