
Overview

This document lists TTCN changes needed to correct problems in the TTCN implementation of PDCP / RRC intra-LTE intercell handover test cases 7.3.5.x and 8.2.4.x which are part of the LTE/SAE test suite ‘iwd-EUTRA-TVB2009-12_D10wk42’.

Corrections 
Change 1 /  Correction to EUTRA_ConfigurationSteps
	Function name
	f_EUTRA_SS_...

	Reason for change
	After changing the HO logic to pre-configure UL grant in the target cell at HO, the UE can come with PDCP data any time (i.e. very quickly which was the intention of this change) after ‘RRCConnectionReconfigurationComplete’ message. The generic HO function does a lot of clean-up after the ‘RRCConnectionReconfigurationComplete’ message (like TA config, RACH procedure config, SRB/DRB reset, etc.) using the CFN flag switched on. So if the UE is quick enough, the UL data can collide with the expected local configuration confirm messages.

	Summary of change
	The configuration procedure was changed to not use CNF messages (i.e. CNF flag is set to false) and thus avoid this race condition and make PDCP Handover more stable.

	TTCN module
	EUTRA_ConfigurationSteps.ttcn 

	MCC160 Comment
	


Before:

	  function f_EUTRA_SS_CommonRadioBearerConfig(CellId_Type p_CellId,
                                              template (value) RadioBearerList_Type p_RadioBearerList,
                                              template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on EUTRA_PTC
  { /* @sic R5s100135 sic@ */
    var boolean v_CnfFlag;
    if (match(valueof(p_TimingInfo), cs_TimingInfo_Now)) {
      v_CnfFlag := tsc_CnfReq;
    } else {
      v_CnfFlag := tsc_NoCnfReq;
    }
    SYS.send(cas_CommonRadioBearerConfig_REQ(p_CellId, p_TimingInfo, p_RadioBearerList, v_CnfFlag));
    if (v_CnfFlag == tsc_CnfReq) {
      SYS.receive(car_CommonRadioBearerConfig_CNF(p_CellId));
    }
}
...

  function f_EUTRA_SS_DrxCtrlConfig(CellId_Type p_CellId,
                                    template (value) DrxCtrl_Type p_NewDrxCtrl,
                                    template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on EUTRA_PTC
  { /* @status    APPROVED */
    var DrxCtrl_Type v_OldDrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_CellId);
    if (ischosen(p_NewDrxCtrl.None) and ischosen(v_OldDrxCtrl.None)) {
      return;
    }
    if (not ischosen(p_NewDrxCtrl.None) and not ischosen(v_OldDrxCtrl.None)) {
      FatalError(__FILE__, __LINE__, "Reconfiguration of DRX - not supported (yet)");
    }
    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_DrxCtrl_REQ(p_CellId, p_TimingInfo, p_NewDrxCtrl));
  }
...

  function f_EUTRA_SS_SRBs_DRBs_Release ( CellId_Type  p_CellId,
                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                          DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def ) runs on EUTRA_PTC
  {
    var template (value) RadioBearerList_Type v_SrbDrbList := {
      cs_SRB_Release(tsc_SRB0),
      cs_SRB_Release(tsc_SRB1),
      cs_SRB_Release(tsc_SRB2)
    };
    var integer i := lengthof(v_SrbDrbList);
    var integer k;
    for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
      v_SrbDrbList[i+k] := cs_DRB_Release(p_DrbList[k]);    /* @sic R5s100279 change 1 sic@ */
    }
    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);
}

...

  function f_EUTRA_SS_SRBs_DRBs_Config ( CellId_Type  p_CellId,
                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                         template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def ) runs on EUTRA_PTC
  {
    var template (value) RadioBearerList_Type v_SrbDrbList := {
      cs_SRB0_ConfigDef,
      cs_SRB1_ConfigDef,
      cs_SRB2_ConfigDef
    };
    var integer i := lengthof(v_SrbDrbList);
    var integer k;
    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {
      v_SrbDrbList[i+k] := p_DrbConfigList[k];          /* @sic R5s100279 change 2 sic@ */
    }
    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo);
 }

...

  function f_EUTRA_SS_ConfigRachProcedure(CellId_Type p_CellId,
                                          template (omit) C_RNTI p_CRNTI,
                                          template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now  // @sic R5s100178 sic@
                                          ) runs on EUTRA_PTC
  {
    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_RachProcedureConfig_REQ(p_CellId, p_CRNTI, p_RachProcedureConfig, p_TimingInfo));
};




After:

	  function f_EUTRA_SS_CommonRadioBearerConfig(CellId_Type p_CellId,
                                              template (value) RadioBearerList_Type p_RadioBearerList,
                                              template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                              boolean p_CnfFlag := tsc_CnfReq) runs on EUTRA_PTC
  { /* @sic R5s100135 sic@ */
    var boolean v_CnfFlag;
    if (match(valueof(p_TimingInfo), cs_TimingInfo_Now) and match(p_CnfFlag, tsc_CnfReq)) {
      v_CnfFlag := tsc_CnfReq;
    } else {
      v_CnfFlag := tsc_NoCnfReq;
    }
    SYS.send(cas_CommonRadioBearerConfig_REQ(p_CellId, p_TimingInfo, p_RadioBearerList, v_CnfFlag));
    if (v_CnfFlag == tsc_CnfReq) {
      SYS.receive(car_CommonRadioBearerConfig_CNF(p_CellId));
    }
}
...

  function f_EUTRA_SS_DrxCtrlConfig(CellId_Type p_CellId,
                                    template (value) DrxCtrl_Type p_NewDrxCtrl,
                                    template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                    boolean p_CnfFlag := tsc_CnfReq) runs on EUTRA_PTC
  { /* @status    APPROVED */
    var DrxCtrl_Type v_OldDrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_CellId);
    if (ischosen(p_NewDrxCtrl.None) and ischosen(v_OldDrxCtrl.None)) {
      return;
    }
    if (not ischosen(p_NewDrxCtrl.None) and not ischosen(v_OldDrxCtrl.None)) {
      FatalError(__FILE__, __LINE__, "Reconfiguration of DRX - not supported (yet)");
    }
    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_DrxCtrl_REQ(p_CellId, p_TimingInfo, p_NewDrxCtrl, p_CnfFlag));
}

...

  function f_EUTRA_SS_SRBs_DRBs_Release ( CellId_Type  p_CellId,
                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                          DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,
                                          boolean p_CnfFlag := tsc_CnfReq ) runs on EUTRA_PTC
  {
    var template (value) RadioBearerList_Type v_SrbDrbList := {
      cs_SRB_Release(tsc_SRB0),
      cs_SRB_Release(tsc_SRB1),
      cs_SRB_Release(tsc_SRB2)
    };
    var integer i := lengthof(v_SrbDrbList);
    var integer k;
    for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
      v_SrbDrbList[i+k] := cs_DRB_Release(p_DrbList[k]);    /* @sic R5s100279 change 1 sic@ */
    }
    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo, p_CnfFlag);
}

...

  function f_EUTRA_SS_SRBs_DRBs_Config ( CellId_Type  p_CellId,
                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                         template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,
                                         boolean p_CnfFlag := tsc_CnfReq ) runs on EUTRA_PTC
  {
    var template (value) RadioBearerList_Type v_SrbDrbList := {
      cs_SRB0_ConfigDef,
      cs_SRB1_ConfigDef,
      cs_SRB2_ConfigDef
    };
    var integer i := lengthof(v_SrbDrbList);
    var integer k;
    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {
      v_SrbDrbList[i+k] := p_DrbConfigList[k];          /* @sic R5s100279 change 2 sic@ */
    }
    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo, p_CnfFlag);
}

...

  function f_EUTRA_SS_ConfigRachProcedure(CellId_Type p_CellId,
                                          template (omit) C_RNTI p_CRNTI,
                                          template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                          template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,  // @sic R5s100178 sic@
                                          boolean p_CnfFlag := tsc_CnfReq ) runs on EUTRA_PTC
  {
    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_RachProcedureConfig_REQ(p_CellId, p_CRNTI, p_RachProcedureConfig, p_TimingInfo, p_CnfFlag));
  };


Change 2 /  Correction to EUTRA_RRCSteps
	Function name
	f_EUTRA_508RRC_IntraLTE_HO_InterCell

	Reason for change
	See change 1 above.

	Summary of change
	See change 1 above.

	TTCN module
	EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before:

	  function f_EUTRA_508RRC_IntraLTE_HO_InterCell ( CellId_Type                           p_SourceCellId,
                                                  CellId_Type                           p_TargetCellId,
                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,
                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,
                                                  template (omit)  MeasConfig           p_MeasConfig    := omit,
                                                  template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,
                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def)  runs on EUTRA_PTC
  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */
    /* @sic R5s100524: common change of HO functions sic@ */
    /* @sic R5-106063: Sequence of inter-cell handover sic@ */
    var DRB_IdentityList_Type v_DRB_IdentityList;
    var integer k;
    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();
    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment
    var Frequency_IE_Type v_Frequency_IE := f_EUTRA_CellInfo_GetFrequencyIEs (p_TargetCellId);
    var CarrierFreqEUTRA v_CarrierFreqEutra := v_Frequency_IE.UL_DL_Earfcn;
    var template (value) MobilityControlInfo v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_TargetCellId, v_CarrierFreqEutra, p_C_RNTI);
    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId);
    var DrxCtrl_Type v_DrxCtrl;
    // Build DRB identity list from inputed radiobearer config list
    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {
      v_DRB_IdentityList[k] := valueof ( p_DrbConfigList[k].Id.Drb );
    }
    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@
    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@
    // Step 3: Configure C-RNTI based contention resolution in target cell
    f_EUTRA_SS_ConfigRachProcedure_HO(p_TargetCellId, cs_TimingInfo_Now, p_C_RNTI);
    // Save new C-RNTI in cell configuration for p_TargetCellId
    f_EUTRA_CellInfo_SetC_RNTI ( p_TargetCellId, p_C_RNTI );
    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_TargetCellId,
                                                               v_Auth_Params,
                                                               v_NextHopChainingCount,
                                                               -,
                                                               v_DRB_IdentityList);
    f_EUTRA_Security_Set ( v_Auth_Params );
    // Step 5: configure DRX at the target cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);
    }
    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@
    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_Def_REQ(p_SourceCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_None));
    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@
    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));
    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE
    f_EUTRA_RRC_ConnReconfigHO_Common(p_SourceCellId,
                                      p_MeasConfig,
                                      v_MobilityControlInfo,
                                      v_RadioResourceConfigDedicated,
                                      cs_508SecurityConfigHO_IntraLTE);
    // Step 9:
    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@
    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@
    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def));
    // Step 11: Tell the SS about the HO    @sic R5w100212 sic @
    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@
    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@
    f_EUTRA_SS_ConfigRachProcedure (p_TargetCellId,
                                    omit,
                                    cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(p_TargetCellId), f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetCellId)));
    // Step 13: Re-establish SRBs and DRB(s) in source cell
    f_EUTRA_SS_SRBs_DRBs_Release ( p_SourceCellId, -, v_DRB_IdentityList );
    f_EUTRA_SS_SRBs_DRBs_Config ( p_SourceCellId, -, p_DrbConfigList );
    // Step 14: Release DRX configuration
    f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, cs_DrxCtrl_None);
  }


After:

	  function f_EUTRA_508RRC_IntraLTE_HO_InterCell ( CellId_Type                           p_SourceCellId,
                                                  CellId_Type                           p_TargetCellId,
                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,
                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,
                                                  template (omit)  MeasConfig           p_MeasConfig    := omit,
                                                  template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,
                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def)  runs on EUTRA_PTC
  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */
    /* @sic R5s100524: common change of HO functions sic@ */
    /* @sic R5-106063: Sequence of inter-cell handover sic@ */
    var DRB_IdentityList_Type v_DRB_IdentityList;
    var integer k;
    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();
    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment
    var Frequency_IE_Type v_Frequency_IE := f_EUTRA_CellInfo_GetFrequencyIEs (p_TargetCellId);
    var CarrierFreqEUTRA v_CarrierFreqEutra := v_Frequency_IE.UL_DL_Earfcn;
    var template (value) MobilityControlInfo v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_TargetCellId, v_CarrierFreqEutra, p_C_RNTI);
    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId);
    var DrxCtrl_Type v_DrxCtrl;
    // Build DRB identity list from inputed radiobearer config list
    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {
      v_DRB_IdentityList[k] := valueof ( p_DrbConfigList[k].Id.Drb );
    }
    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@
    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@
    // Step 3: Configure C-RNTI based contention resolution in target cell
    f_EUTRA_SS_ConfigRachProcedure_HO(p_TargetCellId, cs_TimingInfo_Now, p_C_RNTI);
    // Save new C-RNTI in cell configuration for p_TargetCellId
    f_EUTRA_CellInfo_SetC_RNTI ( p_TargetCellId, p_C_RNTI );
    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_TargetCellId,
                                                               v_Auth_Params,
                                                               v_NextHopChainingCount,
                                                               -,
                                                               v_DRB_IdentityList);
    f_EUTRA_Security_Set ( v_Auth_Params );
    // Step 5: configure DRX at the target cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);
    }
    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@
    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_Def_REQ(p_SourceCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_None));
    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@
    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));
    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE
    f_EUTRA_RRC_ConnReconfigHO_Common(p_SourceCellId,
                                      p_MeasConfig,
                                      v_MobilityControlInfo,
                                      v_RadioResourceConfigDedicated,
                                      cs_508SecurityConfigHO_IntraLTE);
    // Step 9:
    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@
    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@
    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq));
    // Step 11: Tell the SS about the HO    @sic R5w100212 sic @
    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@
    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@
    f_EUTRA_SS_ConfigRachProcedure (p_TargetCellId,
                                    omit,
                                    cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(p_TargetCellId), f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetCellId)),
                                    cs_TimingInfo_Now,
                                    tsc_NoCnfReq);
    // Step 13: Re-establish SRBs and DRB(s) in source cell
    f_EUTRA_SS_SRBs_DRBs_Release ( p_SourceCellId, -, v_DRB_IdentityList, tsc_NoCnfReq );
    f_EUTRA_SS_SRBs_DRBs_Config ( p_SourceCellId, -, p_DrbConfigList, tsc_NoCnfReq );
    // Step 14: Release DRX configuration
    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, cs_TimingInfo_Now, tsc_NoCnfReq);
  }


Change 3 /  Correction to EUTRA_CellCfg_Templates
	Template name
	SYSTEM_CTRL_REQ

	Reason for change
	See change 1 above.

	Summary of change
	See change 1 above.

	TTCN module
	EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	


Before:

	  template (value) SYSTEM_CTRL_REQ cas_RachProcedureConfig_REQ ( CellId_Type p_CellId,
                                                                 template (omit) C_RNTI p_CRNTI,
                                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                                                 template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now  // @sic R5s100178 sic@
                                                                  ) :=
  { /* @status    APPROVED */
    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo),
    Request := {
      Cell  := {
        AddOrReconfigure := {
          Basic    := omit,
          Active   := {
            C_RNTI                := p_CRNTI,
            PhysicalLayerConfigUL := omit,
            RachProcedureConfig   := p_RachProcedureConfig,
            CcchDcchDtchConfig    := omit
          }
        }
      }
    }
};
...

  template (value) SYSTEM_CTRL_REQ cas_DrxCtrl_REQ(CellId_Type p_CellId,
                                                   template (value) TimingInfo_Type p_TimingInfo,
                                                   template (value) DrxCtrl_Type p_DrxCtrl) :=
  { /* @status    APPROVED */
    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo),
    Request := {
      Cell := {
        AddOrReconfigure := {
          Basic := omit,
          Active := {
            C_RNTI := omit,
            PhysicalLayerConfigUL := omit,
            RachProcedureConfig := omit,
            CcchDcchDtchConfig := {
              TimeDomainRestriction := omit,
              DL := omit,
              UL := omit,
              DrxCtrl := p_DrxCtrl,
              TtiBundling := omit
            }
          }
        }
      }
    }
};

...

  template (value) SYSTEM_CTRL_REQ cas_PUCCH_Synch_Config_REQ(CellId_Type p_CellId,
                                                              template (value) TimingInfo_Type p_TimingInfo,
                                                              template (value) PUCCH_Synch_Type p_PUCCH_Synch) :=
    /* ASP to configure PUCCH_Synch */
    /* @status    APPROVED */
    cas_ULGrantAllocation_REQ(p_CellId, p_TimingInfo, omit, p_PUCCH_Synch, omit);



After:

	  template (value) SYSTEM_CTRL_REQ cas_RachProcedureConfig_REQ ( CellId_Type p_CellId,
                                                                 template (omit) C_RNTI p_CRNTI,
                                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                                                 template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,  // @sic R5s100178 sic@
                                                                 boolean p_CnfFlag := tsc_CnfReq ) :=
  { /* @status    APPROVED */
    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo, p_CnfFlag),
    Request := {
      Cell  := {
        AddOrReconfigure := {
          Basic    := omit,
          Active   := {
            C_RNTI                := p_CRNTI,
            PhysicalLayerConfigUL := omit,
            RachProcedureConfig   := p_RachProcedureConfig,
            CcchDcchDtchConfig    := omit
          }
        }
      }
    }
};
...

  template (value) SYSTEM_CTRL_REQ cas_DrxCtrl_REQ(CellId_Type p_CellId,
                                                   template (value) TimingInfo_Type p_TimingInfo,
                                                   template (value) DrxCtrl_Type p_DrxCtrl,
                                                   boolean p_CnfFlag := tsc_CnfReq) :=
  { /* @status    APPROVED */
    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo, p_CnfFlag),
    Request := {
      Cell := {
        AddOrReconfigure := {
          Basic := omit,
          Active := {
            C_RNTI := omit,
            PhysicalLayerConfigUL := omit,
            RachProcedureConfig := omit,
            CcchDcchDtchConfig := {
              TimeDomainRestriction := omit,
              DL := omit,
              UL := omit,
              DrxCtrl := p_DrxCtrl,
              TtiBundling := omit
            }
          }
        }
      }
    }
  };
...

  template (value) SYSTEM_CTRL_REQ cas_PUCCH_Synch_Config_REQ(CellId_Type p_CellId,
                                                              template (value) TimingInfo_Type p_TimingInfo,
                                                              template (value) PUCCH_Synch_Type p_PUCCH_Synch,
                                                              boolean p_CnfFlag := tsc_CnfReq) :=
    /* ASP to configure PUCCH_Synch */
    /* @status    APPROVED */
    cas_ULGrantAllocation_REQ(p_CellId, p_TimingInfo, omit, p_PUCCH_Synch, omit, p_CnfFlag);

